
EXAMPLE 1.4-1: Fuel Element 
A nuclear fuel element consists of a sphere of fissionable material (fuel) with radius rfuel = 5.0 cm 
and kfuel = 1.0 W/m-K that is surrounded by a shell of metal cladding with outer radius rclad = 7.0 
cm and kclad = 300 W/m-K.  The outer surface of the cladding is exposed to helium gas that is 
being heated by the reactor.  The average convection coefficient between the gas and the 
cladding surface is h  = 100 W/m2-K and the temperature of the gas is T∞ = 500ºC.   
 
Inside the fuel element, fission fragments are produced that have high velocities.  The products 
collide with the atoms of the material and provide the thermal energy for the reactor.  This 
process can be modeled as a nonuniform volumetric thermal energy generation.  The volumetric 
generation ( g′′′ ) can be approximated by: 

where 0g′′′= 5x105 W/m3 is the volumetric rate of heat generation at the center of the sphere and 
b = 1.0 is a dimensionless positive constant that characterizes how quickly the generation rate 
decays in the radial direction. 
 
a.) Develop a numerical model for the spherical fuel element using EES.   
 
A function is defined that returns the volumetric generation given the radial position and the 
radius of the fuel element. 
 
function gen(r, r_fuel) 
 "This function defines the volumetric heat generation in the fuel element 
 Inputs: r: radius (m) 
   r_fuel: radius of fuel sphere (m) 
 Output: volumetric heat generation at r (W/m^3)" 
  
 g```_dot_0=5e5 [W/m^3] "volumetric generation rate at the center" 
 b = 1.0 [-] "constant that describes rate of decay" 
 gen=g```_dot_0*exp(-b*r/r_fuel) "volumetric rate of generation" 
end 
 
The next section of the EES code provides the problem inputs. 
               
"EXAMPLE 1.4-1: Fuel Element" 
$UnitSystem SI MASS RAD PA K J 
$Tabstops 0.2 0.4 0.6 3.5 in 
 
"Inputs" 
r_fuel = 5.0 [cm]*convert(cm,m) "fuel radius" 
r_clad = 7.0 [cm]*convert(cm,m) "cladding radius" 
k_fuel = 1.0 [W/m-K] "fuel conductivity" 
k_clad= 300 [W/m-K] "cladding conductivity" 
h_bar = 100 [W/m^2-K] "average convection coefficient" 
T_infinity = converttemp(C,K,500 [C]) "helium temperature"  



 
The numerical solution proceeds by distributing nodes throughout the computational domain that 
stretches from r = 0 to r = rfuel.  There is no reason to include the metal cladding in the numerical 
model.  The cladding increases the thermal resistance that is already present due to convection 
between the external surface of the fuel and the gas; however, this effect can be included using a 
conduction thermal resistance.   
 
The positions of a uniformly distributed set of nodes are obtained from: 

and the distance between adjacent nodes is: 

"Setup nodes" 
N=50 [-]   "number of nodes" 
duplicate i=1,N 
       r[i]=r_fuel*(i-1)/(N-1) "radial position of each node" 
end 
DELTAr=r_fuel/(N-1) "distance between adjacent nodes" 
              
A control volume for an arbitrary internal node is shown in Figure 1. 
 

 
Figure 1: Control volume for an internal node 

 
The energy balances for the internal control volumes are: 

The control volumes are spherical shells so it is appropriate to use a conduction model that is 
consistent with this geometry (see Table 1-2).  Note that this problem could also be solved by 
approximating the inner and outer shells as plane walls with different surface areas, as was done 



in Section 1.4.2.  However, building the proper geometry into the control volume energy 
balances will allow the problem to be solved to a specified accuracy with fewer nodes. 

The temperature differences used to evaluate LHSq  and RHSq  are consistent with the direction of 
the heat transfer shown in Figure 1 whereas the resistance values in the denominators are written 
in the form of 1/rin – 1/rout (e.g., 1/ri-1 – 1/ri) so that the resistances are positive.  The generation 
in each control volume is given by: 

Combining these equations allows the control volume energy balances for the internal nodes to 
be written as: 

"Internal control volume energy balance" 
duplicate i=2,(N-1) 
 4*pi*k_fuel*(T[i-1]-T[i])/(1/r[i-1]-1/r[i])+4*pi*k_fuel*(T[i+1]-T[i])/(1/r[i]-1/r[i+1])+& 
  4*pi*((r[i]+DELTAr/2)^3-(r[i]-DELTAr/2)^3)*gen(r[i],r_fuel)/3=0 
end 
 
The energy balance for the node that is placed at the outer edge of the fuel (i.e., node N) is: 

where Rclad is the resistance to conduction through the cladding: 

and Rconv is the resistance to convection from the surface of the cladding to the gas: 

R_cond_clad=(1/r_fuel-1/r_clad)/(4*pi*k_clad) "conduction resistance of cladding" 
R_conv=1/(4*pi*r_clad^2*h_bar) "convection resistance from surface of cladding" 
4*pi*k_fuel*(T[N-1]-T[N])/(1/r[N-1]-1/r[N])+(T_infinity-T[N])/(R_cond_clad+R_conv)+& 
    4*pi*(r[N]^3-(r[N]-DELTAr/2)^3)*gen(r[N],r_fuel)/3=0 "node N energy balance" 
 
The energy balance for the node placed at the center of the fuel (i.e., node 1) would be: 

4*pi*k_fuel*(T[2]-T[1])/(1/r[1]-1/r[2])+4*pi*((r[1]+DELTAr/2)^3-r[1]^3)*gen(r[1],r_fuel)/3=0 
 "node 1 energy balance" 
 
Executing the EES code will lead to a division by zero error message.  The radial location of 
node 1, r1, is equal to 0 and therefore the 1/r1 term in the denominator of Eq. (4) is infinite.  (The 
actual resistance associated with conducting energy to a point is infinite.)  A similar error will be 
encountered when computing LHSq  for node 2 in Eq. (2).  This problem can be dealt with by 
approximating the conduction between nodes 1 and 2 using a plane wall approximation so that 
the energy balance for node 1 becomes: 

{4*pi*k_fuel*(T[2]-T[1])/(1/r[1]-1/r[2])+4*pi*((r[1]+DELTAr/2)^3-r[1]^3)*gen(r[1],r_fuel)/3=0} 
4*pi*k_fuel*(r[1]+DELTAr/2)^2*(T[2]-T[1])/DELTAr+4*pi*((r[1]+DELTAr/2)^3-r[1]^3)*gen(r[1],r_fuel)/3=0 
 "node 1 energy balance" 
 
The energy balance for node 2 has to be rewritten in the same way: 

"Internal control volume energy balance" 
{duplicate i=2,(N-1) 
 4*pi*k_fuel*(T[i-1]-T[i])/(1/r[i-1]-1/r[i])+4*pi*k_fuel*(T[i+1]-T[i])/(1/r[i]-1/r[i+1])+& 
  4*pi*((r[i]+DELTAr/2)^3-(r[i]-DELTAr/2)^3)*gen(r[i],r_fuel)/3=0 
end} 
duplicate i=3,(N-1) 
 4*pi*k_fuel*(T[i-1]-T[i])/(1/r[i-1]-1/r[i])+4*pi*k_fuel*(T[i+1]-T[i])/(1/r[i]-1/r[i+1])+& 
  4*pi*((r[i]+DELTAr/2)^3-(r[i]-DELTAr/2)^3)*gen(r[i],r_fuel)/3=0 
end 
4*pi*k_fuel*(r[1]+DELTAr/2)^2*(T[1]-T[2])/DELTAr+4*pi*k_fuel*(T[3]-T[2])/(1/r[2]-1/r[3])+& 
 4*pi*((r[2]+DELTAr/2)^3-(r[2]-DELTAr/2)^3)*gen(r[2],r_fuel)/3=0 "node 2 energy balance" 
             
Now, the program will solve.  The solution is converted to Celsius: 
 
duplicate i=1,N 
 T_C[i]=converttemp(K,C,T[i]) "temperature in C" 
end 
 
Figure 2 illustrates the temperature in the fuel element as a function of radius. 
 

 
Figure 2: Temperature distribution within fuel 

 
The maximum temperature in the fuel element is obtained with the max function. 
 
T_max_C=max(T_C[1..N]) "maximum temperature of the fuel, in C" 
 
Figure 3 shows the maximum temperature in the fuel as a function of the number of nodes and 
indicates that the solution converges for more than about 100 nodes. 
 



 
Figure 3: Maximum temperature within fuel as a function of the number of nodes 

 
Several sanity checks can be carried out in order to verify that the solution is physically correct.  
Figure 4 shows the maximum temperature in the fuel as a function of the fuel conductivity and 
for various values of the volumetric generation at the center of the fuel ( 0g′′′ ).  The maximum 
temperature increases as either the fuel conductivity decreases or the generation rate increases. 

Figure 4: Maximum temperature in the fuel as a function of kfuel for various values of 0′′′g . 

 
Finally, we can compare the model with the analytical solution for the limiting case where b=0 
(i.e., the fuel experiences a uniform volumetric generation rate).  The general solution for the 
temperature distribution and temperature gradient within a sphere exposed to a uniform 
generation rate is given in Table 1-3: 

where C1 and C2 are constants of integration.  The temperature at the center of the sphere must 
be bounded and therefore C1 must be equal to 0 by inspection of Eq. (5); alternatively, the 
temperature gradient at the center must be zero which would also require that C1 = 0 according to 
Eq. (6).  The second boundary condition is related to an energy balance at the interface between 
the cladding and the fuel: 

Combining Equations (5) through (7) leads to: 

which can be solved for C2.   
 
"Analytical solution" 
g```_dot=gen(0 [m],r_fuel)           "rate of volumetric generation to use in analytical solution" 
-k_fuel*4*pi*r_fuel^2*(-g```_dot*r_fuel/(3*k_fuel))=& 
      (-g```_dot*r_fuel^2/(6*k_fuel)+C_2-T_infinity)/(R_cond_clad+R_conv)   
     "boundary condition at r=r_fuel" 
 
The analytical solution is obtained at the same radial locations as the numerical solution. 
 
duplicate i=1,N 
 T_an[i]=-g```_dot*r[i]^2/(6*k_fuel)+C_2 "analytical solution" 
 T_an_C[i]=converttemp(K,C,T_an[i]) "in C" 
end 
            
Figure 5 shows the analytical and numerical solutions in the limit that b = 0 for 50 nodes; the 
agreement is nearly exact indicating that the numerical solution is adequate. 

 
Figure 5: Temperature as a function of radius predicted by analytical and numerical models in the limit that 
b = 0. 



 
 


